


## Making Creatures Talk Like NPCs Tutorial

Creatures in Morrowind lack some of the features that NPCs have. To name a few shortcomings, they can only say dialogue that is assigned to their ID, they cannot be assigned to a faction, and they do not have Disposition.



(Notice that Creeper has no topics, no Disposition level, nor the Persuasion option)

What do you do if you want a creature to have these options? You give the player the illusion that they are talking to the creature.



Here we have a minotaur standing next to a pile of editor markers. Those editor markers are actually a custom race, with each body part being an editor marker. This makes them invisible, and unable to be interacted with. When the player interacts with the minotaur, this editor marker NPC does the talking. They are given the same name as the creature to sell the illusion that the player is talking to the creature. A simple OnActivate on the minotaur followed by a NameOfNPC → ForceGreeting would do the trick. However, this is quite limited.



Having the NPC placed in the world like this limits what the creature can do. If the creature were to get too far from its NPC counterpart, it would not be able to tell the NPC counterpart to speak with the player. The solution? Have the creature place the NPC each time it is activated and have the NPC delete itself once it is done speaking. Something like this on the creature:

```
begin GV_TestMinotaur_1_s
short setUpChat
if ( OnActivate == 1 )
    if ( setUpChat == 0 )
        PlaceAtMe, GV_TestMino_Dummy_01, 1,0,0 ;This is the NPC counterpart
        set setUpChat to 1
    endif
endif

if ( setUpChat == 1 )
    set setUpChat to 0
endif

end
```

would be enough to place the NPC each time, but there's a problem. Each time the NPC gets created, their Disposition gets reset. They could appear, you could give them enough money to boost them to 100 Disposition, but the next time you talk with the creature, they are back to 40 (or whatever the default is). The solution is to create a variable on the creature's script that can store the NPC disposition. Then, each time the NPC is created, it can get the saved Disposition from the creature. This is how the script looks with the new variable:

```

begin GV_TestMinotaur_1_s

short setUpChat
short init
short myDisp

if ( init == 0 )
    set myDisp to 70
    set init to 1
endif

if ( OnActivate == 1 )
    if ( setUpChat == 0 )
        PlaceAtMe, GV_TestMino_Dummy_01, 1,0,0
        set GV_TestMino_Dummy_01.transferDisp to myDisp; set the NPCs disposition variable
        set setUpChat to 1
    endif
endif

if ( setUpChat == 1 )
    set setUpChat to 0
endif

end

```

The initial disposition is set in the init “state”. In the script for the NPC, we will have a matching variable, called `transferDisp`. Below is the matching NPC script:

```

begin GV_TestMinoDummy_1_s

float TimePassed
short transferDisp
short getDisp
short state

if ( menuMode == 1 )
    return
endif

if ( state == 0 )
    set TimePassed to 0
    set state to 1
endif

if ( state == 1 )
    set TimePassed to ( TimePassed + GetSecondsPassed )
    if ( TimePassed >= 0.1 )
        set state to 2
    endif
endif

if ( state == 2 )
    ;For some weird reason, disposition is 10 lower each time, so +10 before set
    ;setDisposition must use a local variable, so set the value to a local variable first.
    set getDisp to ( GV_TestMinotaur_01.myDisp + 10 )
    setDisposition, getDisp
    ForceGreeting
    set TimePassed to 0
    set state to 3
endif

if ( state == 3 )
    set TimePassed to (TimePassed + GetSecondsPassed)
    if ( TimePassed >= 0.1 )
        set state to 4
    endif
endif

if ( state == 4 )
    set GV_TestMinotaur_01.myDisp to GetDisposition
    set TimePassed to (TimePassed + GetSecondsPassed)
    if ( TimePassed >= 0.5 )
        set state to 5
    endif
endif

if ( state == 5 )
    disable
    setDelete, 1
endif

end

```

To summarize the above: The NPC appears, sets its Disposition based on the value from the creature, talks to the player, updates the Disposition based on the conversation and passes it to the

creature, and then deletes itself. The states with TimePassed may be excessive, but sometimes the NPC will speak before it gets its Disposition value from the creature, so they are present as a precaution. You can now add specific dialogue to the NPC, or general dialogue to the NPC's race. There is one final piece missing, however: taunting.

When you successfully taunt an NPC, they will attack you. With the above, the NPC will start combat, but then immediately delete themselves. You can speak with the creature again, but they will have no knowledge of the successful taunt. Just like with Disposition, we create a new variable, something like shouldFight. When we successfully taunt the NPC in question, we can set this value in the dialogue:

The screenshot shows a configuration window for an NPC. The 'Speaker Condition' section is expanded, showing a table of conditions. The 'Race' is set to 'GV\_DummyMinotaur'. The 'Function/Variable' table has 'Should Attack' set to '1'. The 'Result' section shows the dialogue 'set shouldFight to 1' and 'Goodbye'.

Function/Variable	Disp
Should Attack	1

(Note that the race is used so that this logic can be shared among all members of the race) We can then add a check for that in the NPC's self-delete state:

```
if ( state == 5 )
    if ( shouldFight == 1 )
        set GV_TestMinotaur_01.isFighting to 1
    endif
    disable
    setDelete, 1
endif
```

and a check for the variable update in the creature's script:

```
if ( isFighting == 1 )
  StartCombat, Player
  set isFighting to 0
endif
```

If you manage to calm the creature and talk with them again, they will be able to be taunted again. With all that, you can now place your creature in the world and let the scripts do the rest:

